
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-EUTRA-B2013-03_D14wk24’.

2. Table of Contents

11. Overview

2. Table of Contents
1
3. Corrections
1
Change 1 – Correction to local function ‘fl_TC_9_2_3_1_20a’
1

3. Corrections

Change 1 – Correction to local function ‘fl_TC_9_2_3_1_20a’
	Function name
	fl_TC_9_2_3_1_20a ()

	Reason for change
	Acc. to TS 24.301 clause 4.4.5:

“… . The UE shall start the ciphering and deciphering of NAS messages when the secure exchange of NAS messages has been established for a NAS signalling connection. From this time onward, unless explicitly defined, the UE shall send all NAS messages ciphered until the NAS signalling connection is released, or the UE performs intersystem handover to A/Gb mode or Iu mode.

The MME shall start ciphering and deciphering of NAS messages as described in subclause 4.4.2.3. From this time onward, except for the SECURITY MODE COMMAND message, the MME shall send all NAS messages ciphered until the NAS signalling connection is released, or the UE performs intersystem handover to A/Gb mode or Iu mode.

Once the encryption of NAS messages has been started between the MME and the UE, the receiver shall discard the unciphered NAS messages which shall have been ciphered according to the rules described in this specification. …”

As network operation with ciphering is configured, the TAU Reject message in step 3 should also be sent ciphered and integrity protected. 

	Summary of change
	TAU Reject message in step 3 modified to be sent ciphered + integrity protected. 

	TTCN module
	LTE_A\9_2\NAS_TrackingArea_EOnly_A.ttcn

	MCC160 comments
	


Before:

	…
  //----------------------------------------------------------------------------
  /*
   * @desc      test body of TC_9_2_3_1_20a
   * @status    APPROVED (LTE_A)
   */
  function fl_TC_9_2_3_1_20a_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var NAS_PlmnId v_HomePlmn:= f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);
    var GutiParameters_Type v_Guti2_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellB);
    var template (omit) MobileIdentity v_MsId := f_GetMSId (NORMAL);
    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL);
    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;
    var template (value) CellPowerList_Type v_CellPowerList;
    var NAS_MSG_Indication_Type v_NasInd;
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var float v_T3346Val := f_EUTRA_SetTimerToleranceMin(eutra_CellB, nasTimer, 300.0); //5minutes @sic R5s140322 sic@
    // ==========================================================================
    //@siclog "Step 1" siclog@
    //+   Set the cell type of cell A to the "Non-Suitable cell".
    //+   Set the cell type of cell B to the "Serving cell"
    v_CellPowerList := {
      cs_CellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ),
      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )
    };
    f_EUTRA_SetCellPowerList ( v_CellPowerList );
    //@siclog "Step 2" siclog@
    //+   The UE transmits a TRACKING AREA UPDATE REQUEST on Cell B.
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,
                                               tsc_RRC_TI_Def,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected ,
                                                                 cr_508_TAU_Request(v_UpdateType,
                                                                                    f_EUTRA_SecurityKSIasme_Get(),
                                                                                    cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                                    v_AdditionalUpdateType)));
    //@siclog "Step 3" siclog@
    //+   The SS transmits a TRACKING AREA UPDATE REJECT message with EMM cause = '' Congestion '' and T3346 IE set to 5 minutes.
    //+   The UE starts timer T3346 with the value provided in the T3346 value IE.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected, // @sic R5-140764 sic@
                                                cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_Cause_Congestion, cs_GPRS_Timer2('5F'O, tsc_GprsTimerUnit_1min, '00101'B))))); // @sic R5s140322 sic@
    //@siclog "Step 4" siclog@
    //+   The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);
    //@siclog "Step 5" siclog@
    //+ Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message before timer T3346 has expired?
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellB, v_T3346Val)) {
      // even though prose says TRACKING AREA UPDATE REQUEST, as soon as UE sends RRC ConnReq, we know something is wrong
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step5: TAU Request received");
    }
    //@siclog "Step 6" siclog@
    //+   Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message after timer T3346 has expired?
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,
                                               tsc_RRC_TI_Def,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                 cr_508_TAU_Request(v_UpdateType,
                                                                                    f_EUTRA_SecurityKSIasme_Get(),
                                                                                    cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                                    v_AdditionalUpdateType)));
    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;
    //@siclog "Step 7" siclog@
    //+   The SS transmits a TRACKING AREA UPDATE ACCEPT message.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_UpdateType,
                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti2_Params),
                                                                  cds_TAIListNonConsecutive_tlv(v_HomePlmn, // Cell B is using the same PLMN ID as Cell A
                                                                                                { '0002'O }), // TAC=2
                                                                  v_EpsBearerCtxtStatus,
                                                                  f_EUTRA_GetCellLAI(eutra_CellB, NORMAL),
                                                                  v_MsId,
                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));
    //@siclog "Step 8" siclog@
    //+   The UE transmits a TRACKING AREA UPDATE COMPLETE message
    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)));
    //At the end of this test procedure sequence, the UE is in end state E-UTRA connected (E2_T3440) according to TS 36.508
  }// function fl_TC_9_2_3_1_20a_Body
…


After:

	…

  //----------------------------------------------------------------------------
  /*
   * @desc      test body of TC_9_2_3_1_20a
   * @status    APPROVED (LTE_A)
   */
  function fl_TC_9_2_3_1_20a_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var NAS_PlmnId v_HomePlmn:= f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);
    var GutiParameters_Type v_Guti2_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellB);
    var template (omit) MobileIdentity v_MsId := f_GetMSId (NORMAL);
    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL);
    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;
    var template (value) CellPowerList_Type v_CellPowerList;
    var NAS_MSG_Indication_Type v_NasInd;
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var float v_T3346Val := f_EUTRA_SetTimerToleranceMin(eutra_CellB, nasTimer, 300.0); //5minutes @sic R5s140322 sic@
    // ==========================================================================
    //@siclog "Step 1" siclog@
    //+   Set the cell type of cell A to the "Non-Suitable cell".
    //+   Set the cell type of cell B to the "Serving cell"
    v_CellPowerList := {
      cs_CellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ),
      cs_CellPower ( eutra_CellB, tsc_ServingCellRS_EPRE )
    };
    f_EUTRA_SetCellPowerList ( v_CellPowerList );
    //@siclog "Step 2" siclog@
    //+   The UE transmits a TRACKING AREA UPDATE REQUEST on Cell B.
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,
                                               tsc_RRC_TI_Def,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected ,
                                                                 cr_508_TAU_Request(v_UpdateType,
                                                                                    f_EUTRA_SecurityKSIasme_Get(),
                                                                                    cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                                    v_AdditionalUpdateType)));
    //@siclog "Step 3" siclog@
    //+   The SS transmits a TRACKING AREA UPDATE REJECT message with EMM cause = '' Congestion '' and T3346 IE set to 5 minutes.
    //+   The UE starts timer T3346 with the value provided in the T3346 value IE.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, // @sic R5-140764 sic@ 
                                                cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_Cause_Congestion, cs_GPRS_Timer2('5F'O, tsc_GprsTimerUnit_1min, '00101'B))))); // @sic R5s140322 sic@
    //@siclog "Step 4" siclog@
    //+   The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);
    //@siclog "Step 5" siclog@
    //+ Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message before timer T3346 has expired?
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellB, v_T3346Val)) {
      // even though prose says TRACKING AREA UPDATE REQUEST, as soon as UE sends RRC ConnReq, we know something is wrong
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step5: TAU Request received");
    }
    //@siclog "Step 6" siclog@
    //+   Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message after timer T3346 has expired?
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,
                                               tsc_RRC_TI_Def,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                 cr_508_TAU_Request(v_UpdateType,
                                                                                    f_EUTRA_SecurityKSIasme_Get(),
                                                                                    cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                                    v_AdditionalUpdateType)));
    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;
    //@siclog "Step 7" siclog@
    //+   The SS transmits a TRACKING AREA UPDATE ACCEPT message.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_UpdateType,
                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti2_Params),
                                                                  cds_TAIListNonConsecutive_tlv(v_HomePlmn, // Cell B is using the same PLMN ID as Cell A
                                                                                                { '0002'O }), // TAC=2
                                                                  v_EpsBearerCtxtStatus,
                                                                  f_EUTRA_GetCellLAI(eutra_CellB, NORMAL),
                                                                  v_MsId,
                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));
    //@siclog "Step 8" siclog@
    //+   The UE transmits a TRACKING AREA UPDATE COMPLETE message
    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)));
    //At the end of this test procedure sequence, the UE is in end state E-UTRA connected (E2_T3440) according to TS 36.508
  }// function fl_TC_9_2_3_1_20a_Body
…


