
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 11.2.8 which is part of the LTE/SAE test suite in the ‘iwd-EUTRA-B2013-03_D14wk10’ ATS delivery.

The test case can be demonstrated to run correctly with one LTE UE (see section 5) on two different test platforms. Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_11_2_8
Test Group:
LTE_IRAT\11\IMS_EmergencyCall_UGC.ttcn
ATS Version:
iwd-EUTRA-B2013-03_D14wk10
System Simulator used:
R&S CMW500



+


Anite Conformance Toolset solution
UE used:
Qualcomm MSM 9x25  
Verification Status:
PASS


4. Corrections required for test case 8.3.1.21
Change 1 – Correction to function ‘f_TC_11_2_8_EUTRA’
	Function name
	f_TC_11_2_8_EUTRA()

	Reason for change
	1.   The current TTCN implementation uses system information combination c4  or c5 depending on the setting of px_RatComb_Tested, as per the prose. However in all other test cases which can be run in either LTE<>UTRAN or LTE<>GERAN paths, system information combination c10 (including both UTRAN and GERAN cell information) is used. If e.g. px_RatCombTested = EUTRA_UTRAN, then the GERAN PTC will be started with function f_GERAN_Init(), but since the system information combination c4 is used, the function fl_EUTRA_NAS_IRAT_Init() will not coordinate with the GERAN PTC. This will cause the function f_GERAN_Init() to block and never terminate, which can cause an incorrect verdict at the end of the test case.
2. 
The initial conditions specified in the prose require the UE to be in state EMM-DEREGISTERED.LIMITED-SERVICE.

	Summary of change
	1.   Initialise the EUTRA cell with system information combination c10 (as issued for other multi-path test cases) regardless of PICS / PIXIT settings.
2.   Include a preamble to reject the UEs initial attach to place it into state EMM-DEREGISTERED.LIMITED-SERVICE.

	TTCN module
	IMS_EmergencyCall_UGC.ttcn

	MCC160 Comment
	


Before change

...

function f_TC_11_2_8_EUTRA() runs on EUTRA_PTC

  {

    // Attach for emergency bearer services / Rejected /

    // No suitable cells in tracking area / Emergency call using the CS domain

    // UTRAN and GERAN only, CDMA2000 covered separately

    var Combination_Type v_Combination;

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      v_Combination := c4;

    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      v_Combination := c5;

    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }
    f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, v_Combination); // @sic R5-131470 sic@

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    // Set neighbour cell to "Suitable Neighbour cell"

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);
    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_SendCoOrd(GERAN, cms_IRAT_Trigger);
    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }

   f_EUTRA_PreambleOtherRAT(eutra_CellA, -, true);
    f_EUTRA_TestBody_Set(true);

    fl_EUTRA_TC_11_2_8_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_ReleaseAllCells();

  } // function f_TC_11_2_8_EUTRA

... 
After change

...

function f_TC_11_2_8_EUTRA() runs on EUTRA_PTC

  {

    // Attach for emergency bearer services / Rejected /

    // No suitable cells in tracking area / Emergency call using the CS domain

    // UTRAN and GERAN only, CDMA2000 covered separately

   /* REMOVED  var Combination_Type v_Combination;

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      v_Combination := c4;

    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      v_Combination := c5;

    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }*/
    f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c10); // @sic R5-131470 sic@

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    // Set neighbour cell to "Suitable Neighbour cell"

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

    f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);
    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

    f_IRAT_SendCoOrd(GERAN, cms_IRAT_Trigger);
    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }

   //REMOVED f_EUTRA_PreambleOtherRAT(eutra_CellA, -, true);

   //Preamble to put UE in EMM-DEREGISTERED.LIMITED-SERVICE 

   f_EUTRA_Preamble_Reject(eutra_CellA, tsc_Cause_TA_NotAllowed, NORMAL); 

    f_EUTRA_TestBody_Set(true);

    fl_EUTRA_TC_11_2_8_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_ReleaseAllCells();

  } // function f_TC_11_2_8_EUTRA
... 
Change 2 – Correction to local function ‘fl_TC_11_2_8_Body’
	Function name
	fl_TC_11_2_8_Body()

	Reason for change
	1.   Switch on of UE not required as UE will be in EMM-DEREGISTERED.LIMITED-SERVICE state (see change 1).
2.   ATTACH REQUEST at step 3 will ve received without integrity protection.
3.   At step 4 the ATTACH REJECT should be sent with the cause value “no SuitableCells in Tracking Area” as per 36.523-1 11.2.8.3.3-3.

4.   ATTACH REQUEST at step 3 should include IMSI in the MobileIdentity field as per 36.523-1 11.2.8.3.3-1.
5.   ATTACH ACCEPT at step 3 should include KSI of ‘000’B (“no key available”) as per 36.523-1 11.2.8.3.3-1.

	Summary of change
	1.   Remove call to f_UT_SwitchOnUE(UT, false).
2.   Use security header type tsc_SHT_NoSecurityProtection at step 3.
3.   Use cause value tsc_Cause_NoSuitableCells at step 4.
4.   Derive value of v_MsId from px_IMSI_def.
5.   Set v_KsiValue to ‘111’B.

	TTCN module
	IMS_EmergencyCall_UGC.ttcn

	MCC160 Comment
	


Before change

...

    function fl_EUTRA_TC_11_2_8_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NAS_PlmnId v_HomePlmn:= f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);

    var NAS_MSG_Indication_Type v_NasInd;

    var template (omit) MobileIdentity v_MsId := f_GetMSId (NORMAL);
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var template (value) TrackingAreaId v_Tai := cr_TAI(v_HomePlmn, tsc_TAC1);

    var NAS_KsiValue v_KsiValue := '000'B;
    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    //@siclog "Step 1" siclog@

    //+ The UE is switched on.

    f_UT_SwitchOnUE(UT, false);
    //@siclog "Step 3" siclog@

   //Steps 1 to 3 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).

    f_UT_RequestIMSEmergencyCall (UT, px_EmergencyCallNumber);

    //+ The UE transmits an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message as specified on Cell A.

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                                tsc_RRC_TI_Def,

                                                cr_EstablishmentCause_emergency_Access,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                   cr_508_ATTACH_REQUEST_Emergency(v_KsiValue, tsc_EAttachType_Emergency, v_MsId, v_AdditionalUpdateType, v_Tai),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,

                                                                                                                  f_GetPdnType())));

    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    //@siclog "Step 4" siclog@

    //+ The SS transmits an ATTACH REJECT message, EMM cause = "Tracking area not allowed".

    //+ (The list of "forbidden tracking areas for roaming" in the UE should now contain TAI-1)

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                   cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_TA_NotAllowed),

                                                                        omit))));

    //@siclog "Step 5" siclog@

    //+ The SS releases the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_SuitableCellRS_EPRE);

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger);

    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }

    //Wait the end of IRAT procedures

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_WaitForCoOrd_Trigger(GERAN);

    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }

  }

... 
After change

...

function fl_EUTRA_TC_11_2_8_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NAS_PlmnId v_HomePlmn:= f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);

    var NAS_MSG_Indication_Type v_NasInd;

    var template (omit) MobileIdentity v_MsId := f_Imsi2MobileIdentity(px_IMSI_Def);
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var template (value) TrackingAreaId v_Tai := cr_TAI(v_HomePlmn, tsc_TAC1);

    var NAS_KsiValue v_KsiValue := '111'B;
    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    //@siclog "Step 1" siclog@

    //+ The UE is switched on.

    //REMOVED f_UT_SwitchOnUE(UT, false);
    //@siclog "Step 3" siclog@

   //Steps 1 to 3 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).

    f_UT_RequestIMSEmergencyCall (UT, px_EmergencyCallNumber);

    //+ The UE transmits an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message as specified on Cell A.

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                                tsc_RRC_TI_Def,

                                                cr_EstablishmentCause_emergency_Access,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                   cr_508_ATTACH_REQUEST_Emergency(v_KsiValue, tsc_EAttachType_Emergency, v_MsId, v_AdditionalUpdateType, v_Tai),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,

                                                                                                                  f_GetPdnType())));

    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    //@siclog "Step 4" siclog@

    //+ The SS transmits an ATTACH REJECT message, EMM cause = "Tracking area not allowed".

    //+ (The list of "forbidden tracking areas for roaming" in the UE should now contain TAI-1)

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                   cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_TA_NotAllowed),

                                                                        omit))));

    //@siclog "Step 5" siclog@

    //+ The SS releases the RRC connection.

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_SuitableCellRS_EPRE);

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger);

    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }

    //Wait the end of IRAT procedures

    if(pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {

      f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    }

    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {

      f_IRAT_WaitForCoOrd_Trigger(GERAN);

    }

    else  {

      FatalError (__FILE__, __LINE__, "Invalid RAT combination configured");

    }

  }

... 
Change 3 – Correction to template ‘cr_508ATTACH_REQUEST_Emergency’
	Template name
	cr_508ATTACH_REQUEST_Emergency()

	Reason for change
	1.   LastVisitedRegisteredTAI can be omitted according to 36.508 Table 4.7.2-4.
2.   msNetworkFeatureSupport can be included according to 24.301 cl. 8.2.4.16.
3.   The implementation of the template cr_UENetworkCap_Emergencywill only allow UEs to indicate support for ONLY the NULL algorithms EEA0 and EIA0. Further, there is no requirement in the prose to check support of the NULL algorithms, so the template cr_UENetworkCap_Any can be used instead.

	Summary of change
	1.   Add “ifpresent” to LastVisitedRegisteredTAI field.
2.   Allow msNetworkFeatureSupport to be included.
3.   Use template cr_UENetworkCap_Any instead of cr_UENetworkCap_Emergency.

	TTCN module
	IMS_EmergencyCall_UGC.ttcn

	MCC160 Comment
	


Before change

...

template (present) UE_NetworkCap cr_UENetworkCap_Emergency(template IEI8_Type p_IEI) :=

  { // 24.301 cl. 9.9.3.34

    iei := p_IEI,                     /* present in case of TLV; omit in case of LV */

    iel := ?,                         /* present in case of LV or TLV; omit in case of V */

    eeaCap := '80'O, //EEA0

    eiaCap := '80'O, //EIA0
    ueaCap := *,

    uiaCap := *,

    srvccCap := *,

    spare := *

  };

template (present) NAS_UL_Message_Type cr_508_ATTACH_REQUEST_Emergency(template (present) NAS_KsiValue p_KsiValue,

                                                                         template (present) NAS_AttDetValue_Type p_AttachTypeValue,

                                                                         template MobileIdentity p_MobilityId,

                                                                         template AdditionalUpdateType p_AdditionalUpdateType,

                                                                         template TrackingAreaId p_LastVRTai) :=

  { //24.301 cl. 8.2.4

    aTTACH_REQUEST := {

      securityHeaderType       := tsc_SHT_NoSecurityProtection,

      protocolDiscriminator    := tsc_PD_EMM,

      messageType              := tsc_MT_AttachRequest,

      nasKeySetId              := cr_NAS_KeySetIdentifier(p_KsiValue,'0'B), // Native securtity context

      epsAttachType            := cr_EPS_AttachType(p_AttachTypeValue),

      epsMobileIdentity        := p_MobilityId,

      ueNetworkCapability      := cr_UENetworkCap_Emergency(?),

      esmMessage               := cr_ESM_MessageContainer,

      oldPtmsiSignature        := cr_PTMSI_SignatureAny ifpresent,

      additionalGuti           := cr_MobileIdAny ('50'O) ifpresent,

      lastVisitedRegisteredTai := p_LastVRTai,

      drxParameter             := cr_DRXparameter_Any ('5C'O) ifpresent,

      msNetworkCapability      := cr_MS_NetworkCap_Any ifpresent,

      oldLai                   := omit,

      tmsiStatus               := omit,

      msClassmark2             := cr_MS_Clsmk2_Any_tlv ifpresent,

      msClassmark3             := cr_MSCLSMK3_Any ifpresent,

      supportedCodecList       := cr_CodecListAny ifpresent,

      addUpdateType            := p_AdditionalUpdateType,

      voiceDomainPref          := cr_VoiceDomainPref_Any ifpresent,

      deviceProperties         := omit,

      oldGUTI_Type             := omit,

      msNetworkFeatureSupport  := omit,

      tmsiBasedNRIContainer    := cr_NwkResourceIdAny ifpresent // @sic R5s130195 Baseline Moving sic@

    }

  };  
... 
After change

...

/* REMOVED template (present) UE_NetworkCap cr_UENetworkCap_Emergency(template IEI8_Type p_IEI) :=

  { // 24.301 cl. 9.9.3.34

    iei := p_IEI,                     /* present in case of TLV; omit in case of LV */

    iel := ?,                         /* present in case of LV or TLV; omit in case of V */

    eeaCap := '80'O, //EEA0

    eiaCap := '80'O, //EIA0

    ueaCap := *,

    uiaCap := *,

    srvccCap := *,

    spare := *

  };*/
template (present) NAS_UL_Message_Type cr_508_ATTACH_REQUEST_Emergency(template (present) NAS_KsiValue p_KsiValue,

                                                                         template (present) NAS_AttDetValue_Type p_AttachTypeValue,

                                                                         template MobileIdentity p_MobilityId,

                                                                         template AdditionalUpdateType p_AdditionalUpdateType,

                                                                         template TrackingAreaId p_LastVRTai) :=

  { //24.301 cl. 8.2.4

    aTTACH_REQUEST := {

      securityHeaderType       := tsc_SHT_NoSecurityProtection,

      protocolDiscriminator    := tsc_PD_EMM,

      messageType              := tsc_MT_AttachRequest,

      nasKeySetId              := cr_NAS_KeySetIdentifier(p_KsiValue,'0'B), // Native securtity context

      epsAttachType            := cr_EPS_AttachType(p_AttachTypeValue),

      epsMobileIdentity        := p_MobilityId,

      ueNetworkCapability      := cr_UENetworkCap_Any(omit),

      esmMessage               := cr_ESM_MessageContainer,

      oldPtmsiSignature        := cr_PTMSI_SignatureAny ifpresent,

      additionalGuti           := cr_MobileIdAny ('50'O) ifpresent,

      lastVisitedRegisteredTai := p_LastVRTai ifpresent,

      drxParameter             := cr_DRXparameter_Any ('5C'O) ifpresent,

      msNetworkCapability      := cr_MS_NetworkCap_Any ifpresent,

      oldLai                   := omit,

      tmsiStatus               := omit,

      msClassmark2             := cr_MS_Clsmk2_Any_tlv ifpresent,

      msClassmark3             := cr_MSCLSMK3_Any ifpresent,

      supportedCodecList       := cr_CodecListAny ifpresent,

      addUpdateType            := p_AdditionalUpdateType,

      voiceDomainPref          := cr_VoiceDomainPref_Any ifpresent,

      deviceProperties         := omit,

      oldGUTI_Type             := omit,

      msNetworkFeatureSupport  := cr_MS_NetworkFeatureSupport('C'H) ifpresent,

      tmsiBasedNRIContainer    := cr_NwkResourceIdAny ifpresent // @sic R5s130195 Baseline Moving sic@

    }

  };
... 
Change 4 – Correction to function ‘f_TC_11_2_8_UTRAN’
	Function name
	f_TC_11_2_8_UTRAN()

	Reason for change
	1.   System information on cell 5 is currently not being configured.
2.   Call to function f_UTRAN_Postamble expects idle/connected mode RAU which will not happen in this case.

	Summary of change
	1.   Configure default system information on cell 5.
2.   Function f_UTRAN_Postamble replaced by a dedicate sequence performing CC Disconnect, RRC Connection Release and UTRAN Cell Release steps.

	TTCN module
	IMS_EmergencyCall_UGC_UTRAN.ttcn

	MCC160 Comment
	


Before change

...

function f_TC_11_2_8_UTRAN() runs on UTRAN_PTC

  {

    //Initialise test case

    f_UTRAN_Init(EUTRA_UTRAN);

    //Create UTRAN cell5

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    //First Trigger

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Steps 4a1-4a14 of 11.2.8 UTRAN" siclog@

    f_UTRAN_EmgCallSetup(utran_Cell5);

    f_UTRAN_TestBody_Set(false);

    // send coordination message for end of test case

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    // Swith off/power off and release cells

    f_UTRAN_Postamble(utran_Cell5, U2_CONNECTED);
  }  
... 
After change

...

function f_TC_11_2_8_UTRAN() runs on UTRAN_PTC

  {

    //Initialise test case

    f_UTRAN_Init(EUTRA_UTRAN);

    //Create UTRAN cell5

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

  //Start sending System Information

    f_UTRAN_SendDefSysInfo ( utran_Cell5 );
    //First Trigger

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_UTRAN_TestBody_Set(true);

    //@siclog "Steps 4a1-4a14 of 11.2.8 UTRAN" siclog@

    f_UTRAN_EmgCallSetup(utran_Cell5);

    f_UTRAN_TestBody_Set(false);

    // send coordination message for end of test case

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    //CC Disconnect followed by RRC connection release followed by release of the UTRAN cell resources

  f_UTRAN_CC_Disconnect(utran_Cell5, cs_TI_MO); 

  f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Dch, true); 

  v_CellConfig := f_UTRAN_CellInfo_GetConfigType(utran_Cell5);  

  //Release the UTRAN cell

  f_UTRAN_ReleaseCell(utran_Cell5, v_CellConfig);
  }
... 
Change 5 – Correction to function ‘f_UTRAN_EmgCallSetup’
	Function name
	f_UTRAN_EmgCallSetup()

	Reason for change
	After Authentication procedure, the network should trigger a security mode procedure for CS-domain, this also acts as an acceptance for CM service request (see 24.008 sec 4.5.1.1).

This change requires a Prose CR 
(see attached “draft_R5-144xxx_TS-36523-1_TC-11-2-8.zip”)

	Summary of change
	Added security mode command procedure after authentication procedure steps and replaced the call to standard function f_UTRAN_M0_SpeechCall_Steps11_16 with a call to new dedicated function f_UTRAN_EmgCallSetup_Steps11_16.

	TTCN module
	Common\UTRAN\UTRAN_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

...

  /*

   * @desc      This procedure is used in IMS Emg Call test cases (derived from  test case 11.2.8 UTRAN)

   * @param     p_CellId

   * @status

   */

  function f_UTRAN_EmgCallSetup(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC

  {

    var RLC_TR_DATA_IND v_RLC_TR_DATA_IND;

    var InitialUE_Identity v_InitialUE_Id;

    var RRC_DATA_IND v_RRC_DataInd;

    var template (present) EstablishmentCause v_EstCause := emergencyCall;

    var START_Value v_START_Value := '00000000000000000000'B;

    var KeySeq v_KeySeq := '111'B; //cipher key sequence number IE: no key is available

    var float v_Wait := 60.0; // 60 seconds to allow UE to complete the procedure on CDMA2000 cell

    var TI v_TI_R;

    timer t_Wait := v_Wait ; // Local wait Timer

    t_Wait.start;

    alt {

      //Step 2

      //Check: Does the UE transmit an RRC CONNECTION REQUEST message on Cell 5 with Establishment cause: Emergency Call in the next [60] seconds?

      [] U_TM.receive(car_RRC_ConnReq(p_CellId,

                                      tsc_RB0,

                                      cr_108_RRC_ConnReq_r8(v_EstCause))) -> value v_RLC_TR_DATA_IND {

        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 2");

        v_InitialUE_Id := v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

        f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);

      }

      [] t_Wait.timeout {

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 2: RRC Connection Request not received within wait time");

      }

    }

    //Step 3

    // The SS transmits an RRC CONNECTION SETUP message.

    //Step 4

    // The UE transmits an RRC CONNECTION SETUP COMPLETE message.

    f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq(p_CellId,v_InitialUE_Id);

    //Step 5

    // Check: Does the UE transmit a CM SERVICE REQUEST with CM service type IE indicating "Emergency call establishment"?

    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                        tsc_RB3,

                                        cr_U_ServReq_MO_Emergency(v_KeySeq, ?))) -> value v_RRC_DataInd; // UE id to be clarified

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 5");

    v_START_Value := v_RRC_DataInd.ttcn_start;

    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);

    //Step 6-7

    // The SS transmits an AUTHENTICATION REQUEST message.

    // The UE transmits an AUTHENTICATION RESPONSE message.

    f_UTRAN_MM_Authentication();

    //Step 8

    // Check: Does the UE transmit an EMERGENCY SETUP message?

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_ESetup(cr_BcapSpeechMO))) -> value v_RRC_DataInd;

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 8");

    //Steps 9-14 of 11.2.8 UTRAN

    // steps 11-16 of 34.108 cl. 7.2.3.2.3

    v_TI_R := v_RRC_DataInd.msg.esetup.ti;

    f_UTRAN_M0_SpeechCall_Steps11_16(p_CellId, v_TI_R); // @sic R5-142954 sic@
  }  // f_UTRAN_EmgCallSetup
... 
After change

...

  /*

   * @desc      This procedure is used in IMS Emg Call test cases (derived from  test case 11.2.8 UTRAN)

   * @param     p_CellId

   * @status

   */

  function f_UTRAN_EmgCallSetup(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC

  {

    var RLC_TR_DATA_IND v_RLC_TR_DATA_IND;

    var InitialUE_Identity v_InitialUE_Id;

    var RRC_DATA_IND v_RRC_DataInd;

    var template (present) EstablishmentCause v_EstCause := emergencyCall;

    var START_Value v_START_Value := '00000000000000000000'B;

    var KeySeq v_KeySeq := '111'B; //cipher key sequence number IE: no key is available

    var float v_Wait := 60.0; // 60 seconds to allow UE to complete the procedure on CDMA2000 cell

    var TI v_TI_R;

    timer t_Wait := v_Wait ; // Local wait Timer

    t_Wait.start;

    alt {

      //Step 2

      //Check: Does the UE transmit an RRC CONNECTION REQUEST message on Cell 5 with Establishment cause: Emergency Call in the next [60] seconds?

      [] U_TM.receive(car_RRC_ConnReq(p_CellId,

                                      tsc_RB0,

                                      cr_108_RRC_ConnReq_r8(v_EstCause))) -> value v_RLC_TR_DATA_IND {

        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 2");

        v_InitialUE_Id := v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

        f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);

      }

      [] t_Wait.timeout {

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 2: RRC Connection Request not received within wait time");

      }

    }

    //Step 3

    // The SS transmits an RRC CONNECTION SETUP message.

    //Step 4

    // The UE transmits an RRC CONNECTION SETUP COMPLETE message.

    f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq(p_CellId,v_InitialUE_Id);

    //Step 5

    // Check: Does the UE transmit a CM SERVICE REQUEST with CM service type IE indicating "Emergency call establishment"?

    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                        tsc_RB3,

                                        cr_U_ServReq_MO_Emergency(v_KeySeq, ?))) -> value v_RRC_DataInd; // UE id to be clarified

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 5");

    v_START_Value := v_RRC_DataInd.ttcn_start;

    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);

    //Step 6-7

    // The SS transmits an AUTHENTICATION REQUEST message.

    // The UE transmits an AUTHENTICATION RESPONSE message.

    f_UTRAN_MM_Authentication();

    //Security Mode procedure

    f_UTRAN_RRC_Security(p_CellId, true, cs_domain);
    //Step 8

    // Check: Does the UE transmit an EMERGENCY SETUP message?

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_ESetup(cr_BcapSpeechMO))) -> value v_RRC_DataInd;

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "f_UTRAN_EmgCallSetup Step 8");

    //Steps 9-14 of 11.2.8 UTRAN

    // steps 11-16 of 34.108 cl. 7.2.3.2.3

    v_TI_R := v_RRC_DataInd.msg.esetup.ti;

    f_UTRAN_EmgCallSetup_Steps11_16(p_CellId, v_TI_R); 

  }  // f_UTRAN_EmgCallSetup
... 
  /*

   * @desc      This procedure is used in IMS Emg Call test cases (derived from  test case 11.2.8 UTRAN)

   * @param     p_CellId

   * @status

   */

  function f_UTRAN_EmgCallSetup_Steps11_16(UTRAN_CellId_Type p_CellId,

                                           TI p_TI_R) runs on UTRAN_PTC

  {

    var TI v_TI_R := p_TI_R;

    var TI v_TI_S := v_TI_R;

    v_TI_S.tiFlag := '1'B;

    //Steps 11 to 16 of the generic test procedure in TS 34.108 subclause 7.2.3.2.3 are performed on Cell 5.

    //Step 9

    // The SS transmits a Call Proceeding message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_CallProc(v_TI_S)));

    //Step 10-11

    // The SS transmits a Radio Bearer Setup message and receives a Radio Bearer Setup Complete message

    f_UTRAN_RB_SetUp_Speech(p_CellId);

    //Step 12

    // The SS transmits an Alerting message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_Alert(v_TI_S)));

    //Step 13

    // The SS transmits a Connect message

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_Connect(v_TI_S)));

    //Step 14

    //Receive Connect Acknowledge message

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_ConnAck(v_TI_R)));

  }  // f_UTRAN_EmgCallSetup_Steps11_16
... 
5. Execution Log Files

5.1 Qualcomm MSM 9x25
The Qualcomm MSM 9x25 UE passed this test case on R&S CMW500 and Anite Conformance Toolset solution in LTE FDD band 4 and UTRA FDD band 2. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file: 
Rohde-Schwarz\tc_11_2_8_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

· Test Case Execution log file: 
Anite\ TC_11_2_8_20140613_102013.htm   
(Note: PICS/PIXIT settings are captured in separate file TC_11_2_8_PIXIT.xml)
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

